

Gestione Impresa

Un'impresa di produzione produce un solo tipo di merce. Ci sono 40 operai, ciascuno dei quali produce 20 unità di merce al mese. La domanda fluttua nel corso di un semestre secondo la seguente tabella:

Mese	1	2	3	4	5	6
Unità richieste	700	600	500	800	900	800

Per adattare la produzione alle fluttuazioni, si impiega lavoro straordinario (ogni operaio può produrre al massimo 6 unità in più al mese, al costo unitario di 5 euro), si conserva merce in magazzino (l'uso del magazzino comporta un costo di 10 euro al mese per unità di merce), si assume o licenzia personale (il numero di operai può variare al massimo di 5 unità in più o in meno ogni mese, con un sovrapprezzo di 500 euro per ogni assunzione e 700 euro per ogni licenziamento).

All'inizio del periodo di gestione il magazzino è vuoto, e si richiede che debba essere vuoto anche alla fine del semestre. Si vuole pianificare la gestione dell'impresa in modo da massimizzare i guadagni. Risolvere il problema presentando una formulazione matematica, un modello di AMPL con il relativo file di dati, e la soluzione ottenuta con il solutore CPLEX. Qual è il valore della funzione obiettivo se si risolve un rilassamento continuo del problema?

Soluzione

FORMULAZIONE.

- *Indici:* sia i un indice sull'insieme $\{1, \dots, N\}$ dei mesi nell'orizzonte temporale ($N = 6$).
- *Parametri:*
 - sia F il numero iniziale di operai nella ditta ($F = 40$);
 - sia P il numero di unità di merce prodotte da ogni operaio con lavoro ordinario ($P = 20$);
 - sia r_i la richiesta all' i -esimo mese (si veda la tabella nel testo del problema);
 - sia U_s il numero massimo di unità producibili per operaio durante il lavoro straordinario ($U_s = 6$);
 - sia C_s il costo unitario per unità prodotta con il lavoro straordinario ($C_s = 5$);
 - sia C_m il costo unitario mensile di stoccaggio in magazzino ($C_m = 10$);
 - sia V la variazione massima di operai (in più o in meno) al mese, in conseguenza di assunzioni / licenziamenti ($V = 5$);
 - sia C_a il costo fisso di un'assunzione ($C_a = 500$);
 - sia C_l il costo fisso di un licenziamento ($C_l = 700$);
- *Variabili:*
 - sia x_i il numero di operai impiegati all' i -esimo mese;
 - sia y_i la quantità di merce prodotta con lavoro straordinario all' i -esimo mese;
 - sia z_i la quantità di merce in magazzino all' i -esimo mese;
 - sia w_i il denaro speso per assunzioni e licenziamenti all' i -esimo mese.

- *Funzione obiettivo:*

$$\min \sum_{i=1}^N (C_s y_i + C_m z_i + w_i)$$

- *Vincoli:*
 1. (produzione) per ogni $i \leq N$: $y_i \leq U_s x_i$;
 2. (assunzioni) per ogni $i \leq N$: $x_i - x_{i-1} \leq V$;
 3. (licenziamenti) per ogni $i \leq N$: $x_{i-1} - x_i \leq V$;
 4. (bilancio magazzino) per ogni $i \leq N$: $Px_i + z_{i-1} + y_i = r_i + z_i$;
 5. (politica assunzioni) per ogni $i \leq N$: $w_i \geq C_a(x_i - x_{i-1})$;
 6. (politica licenziamenti) per ogni $i \leq N$: $w_i \geq C_l(x_{i-1} - x_i)$;

MODELLO AMPL

```
# gestioneimpresa.mod - modello AMPL per il problema di gestione impresa
```

```
set MESI;
param domanda {MESI};
param operai_inizio;
param produzione_operai;
param max_straord;
param costo_straord;
param costo_magazz;
param max_assunti;
param max_licenz;
param costo_assunzione;
param costo_licenz;

var operai {{0} union MESI} >= 0, integer;
var prod_straord {MESI} >= 0;
var magazzino {{0} union MESI} >= 0;
var politica {MESI} >= 0;

minimize costo : sum {i in MESI} (costo_straord * prod_straord[i] +
                                   costo_magazz * magazzino[i] + politica[i]);

subject to produzione {i in MESI} :
    prod_straord[i] <= max_straord * operai[i];

subject to assunzioni {i in MESI} :
    operai[i] - operai[i - 1] <= max_assunti;

subject to licenziamenti {i in MESI} :
    operai[i - 1] - operai[i] <= max_licenz;

subject to bilancio_magazz {i in MESI} :
    produzione_operai * operai[i] + magazzino[i - 1] +
    prod_straord[i] = domanda[i] + magazzino[i];

subject to politica_assunzioni {i in MESI} :
    politica[i] >= costo_assunzione * (operai[i] - operai[i - 1]);

subject to politica_licenz {i in MESI} :
    politica[i] >= costo_licenz * (operai[i - 1] - operai[i]);
```

DATI AMPL

```
# gestioneimpresa.dat - dati AMPL per il problema di gestione impresa

set MESI := 1 2 3 4 5 6 ;
param domanda :=
    1 700
    2 600
    3 500
    4 800
    5 900
    6 800 ;
param operai_inizio := 40;
param produzione_operai := 20;
param max_straord := 6;
param costo_straord := 5;
param costo_magazz := 10;
param max_assunti := 5;
param max_licenz := 5;
param costo_assunzione := 500;
param costo_licenz := 700;

let magazzino[0] := 0;
let operai[0] := operai_inizio;
fix magazzino[0];
fix operai[0];
```

SOLUZIONE

```
ILOG CPLEX 8.100, licensed to "politecnico-milano", options: e m b q
CPLEX 8.1.0: optimal integer solution; objective 10000
9 MIP simplex iterations
0 branch-and-bound nodes
costo = 10000
```

```
operai [*] :=
```

```
0 40
```

```
1 35
```

```
2 33
```

```
3 33
```

```
4 33
```

```
5 33
```

```
6 33
```

```
;
```

```
prod_straord [*] :=
```

```
1 0
```

```
2 0
```

```
3 0
```

```
4 0
```

```
5 160
```

```
6 140
```

```
;
```

```
magazzino [*] :=
```

```
0 0
```

```
1 0
```

```
2 60
```

```
3 220
```

```
4 80
```

```
5 0
```

```
6 0
```

```
;
```

```
politica [*] :=
```

```
1 3500
```

```
2 1400
```

```
3 0
```

```
4 0
```

```
5 0
```

```
6 0
```

```
;
```